Variable internal stores (1 nutrient) + fluctuating temperature affecting μ∞

In[2]:=
```
<< EcoEvo`
```

Out[2]=
```
EcoEvo Package Version 1.7.3 (October 4, 2023)
Christopher A. Klausmeier <christopher.klausmeier@gmail.com>
```

In[59]:=
```
SetModel[{
   Aux[R] → {Equation :→ a (Rin - R) - v[R] n, Color → Blue},
   Pop[pop] → {
     Component[Q] :→ {Equation :→ v[R] - μ[Q] Q, Type → "Intensive", Color → Orange},
     Component[n] :→ {Equation :→ (μ[Q] - m) n, Color → Darker@Green}
    },
   Parameters :→ {a > 0, Rin > 0, vmax > 0, H > 0, μ∞ > 0, Qmin > 0, m > 0},
   Period :→ τ
  }]
```

In[60]:=
```
v[R_] := vmax R / (R + H);
μ[Q_] := μ∞ (1 - Qmin / Q);

a = 1;
Rin = 10;
vmax = 10;
H = 1;
μ∞ := E^T; (* max growth rate is an exponential function of temperature *)
Qmin = 1;
m = 0.5;
```

In[83]:=
```
T := Tav + A Sin[2 π t / τ]; (* temperature is a sinusoidal function of time *)
```

In[84]:=
```
Tav = 0;
A = 2;
τ = 1;
```

Dynamics starting with low abundance goes through two phases.

In[90]:=
```
sol = EcoSim[{R → Rin, Q → Qmin, n → 0.0001}, 25];
PlotDynamics[sol]
PlotDynamics[sol, {n}, Logged → True]
```

Out[91]=



Out[92]=



The variables fluctuate over each period, but the average trend of the population is exponential growth (positive invasion fitness). To calculate the invasion rate, we solve for the quota dynamics with no population and R=Rin:

In[98]:=
```
qsol = EcoSim[{R → Rin, Q → Qmin, n → 0}, 10 τ];
PlotDynamics[qsol, {Q}]
```

Out[99]=



The quota reaches a stable cycle, so let's find it exactly:

In[100]:=

```
qec = FindEcoCycle[FinalSlice[qsol]];
PlotDynamics[qec, {Q}]
Plot[T, {t, 0, τ}, PlotStyle → Red, AxesLabel → {"t", Style["T", Red]}]
```

Infinity: Indeterminate expression −∞ + ∞ encountered.

Infinity: Indeterminate expression −∞ + ∞ encountered.

Out[101]=



Out[102]=



Plot instantaneous growth rate (1/N)dN/dt forced by the quota dynamics:

In[109]:=

```
Plot[μ[Q[t]] - m /. qec, {t, 0, τ}, Filling → Axis]
```

Out[109]=



and integrate (1/N)dN/dt over one period to calculate the invasion rate:

In[107]:=

```
inv = NIntegrate[μ[Q[t]] - m /. qec, {t, 0, τ}] / τ
```

Out[107]=

```
1.28826
```

Comparing the invasion dynamics with our calculated rate:

In[108]:=

```
Show[
 PlotDynamics[sol, {n}, Logged → True],
 LogPlot[10^(-3.5) E^(inv t), {t, 0, 10}, PlotStyle → Dashed]
]
```

Out[108]=



for slides