


```

In[*]:= (* requires EcoEvo package <https://
github.com/cklausme/EcoEvo> -- run this once (ever) to install *)
PacletInstall[
  "https://github.com/cklausme/EcoEvo/raw/master/Paclets/EcoEvo-1.7.2.paclet"]
Out[*]:=
PacletObject[  Name: EcoEvo
Version: 1.7.2 ]

```

## Set model

```

In[*]:= << EcoEvo`
Out[*]:=
EcoEvo Package Version 1.7.2 (September 1, 2023)
Christopher A. Klausmeier <christopher.klausmeier@gmail.com>

In[*]:= SetModel[{
  Pop[n1] → {Equation ⇒ (r1 - α11 n1 - α12 n2) n1},
  Pop[n2] → {Equation ⇒ (r2 - α21 n1 - α22 n2) n2},
  Parameters ⇒ {r1 > 0, r2 > 0, α11 > 0, α12 > 0, α21 > 0, α22 > 0}
}];

```

## Analysis

Solve for equilibria:

```

In[*]:= eq = SolveEcoEq[]
Out[*]:=
{ {n1 → 0, n2 → 0}, {n1 →  $\frac{r1}{\alpha11}$ , n2 → 0}, {n1 → 0, n2 →  $\frac{r2}{\alpha22}$ },
  {n1 →  $-\frac{-r2 \alpha12 + r1 \alpha22}{\alpha12 \alpha21 - \alpha11 \alpha22}$ , n2 →  $-\frac{r2 \alpha11 - r1 \alpha21}{\alpha12 \alpha21 - \alpha11 \alpha22}$ } }

```

Jacobian matrix:

```

In[*]:= EcoJacobian[] // MatrixForm
Out[*]//MatrixForm=

$$\begin{pmatrix} r1 - 2 n1 \alpha11 - n2 \alpha12 & -n1 \alpha12 \\ -n2 \alpha21 & r2 - n1 \alpha21 - 2 n2 \alpha22 \end{pmatrix}$$


```

Calculate eigenvalues at each equilibrium:

```
In[*]:= EcoEigenvalues[eq[[1]]]
```

```
Out[*]=
```

$$\{r1, r2\}$$

```
In[*]:= Simplify[EcoEigenvalues[eq[[2]]]]
```

```
Out[*]=
```

$$\left\{-r1, r2 - \frac{r1 \alpha_{21}}{\alpha_{11}}\right\}$$

```
In[*]:= Simplify[EcoEigenvalues[eq[[3]]]]
```

```
Out[*]=
```

$$\left\{-r2, r1 - \frac{r2 \alpha_{12}}{\alpha_{22}}\right\}$$

```
In[*]:= Simplify[EcoEigenvalues[eq[[4]]]]
```

```
Out[*]=
```

$$\left\{-\frac{1}{2 \alpha_{12} \alpha_{21} - 2 \alpha_{11} \alpha_{22}} \left( r2 \alpha_{11} \alpha_{12} - r1 \alpha_{11} \alpha_{22} - r2 \alpha_{11} \alpha_{22} + r1 \alpha_{21} \alpha_{22} + \sqrt{4 (r2 \alpha_{11} - r1 \alpha_{21}) (r2 \alpha_{12} - r1 \alpha_{22}) (-\alpha_{12} \alpha_{21} + \alpha_{11} \alpha_{22}) + (r2 \alpha_{11} (\alpha_{12} - \alpha_{22}) + r1 (-\alpha_{11} + \alpha_{21}) \alpha_{22})^2} \right), \right. \\ \left. \frac{1}{2 \alpha_{12} \alpha_{21} - 2 \alpha_{11} \alpha_{22}} \left( -r2 \alpha_{11} \alpha_{12} + r1 \alpha_{11} \alpha_{22} + r2 \alpha_{11} \alpha_{22} - r1 \alpha_{21} \alpha_{22} + \sqrt{4 (r2 \alpha_{11} - r1 \alpha_{21}) (r2 \alpha_{12} - r1 \alpha_{22}) (-\alpha_{12} \alpha_{21} + \alpha_{11} \alpha_{22}) + (r2 \alpha_{11} (\alpha_{12} - \alpha_{22}) + r1 (-\alpha_{11} + \alpha_{21}) \alpha_{22})^2} \right) \right\}$$

Stability conditions:

```
In[*]:= EcoStableQ[eq[[1]]]
```

```
Out[*]=
```

False

```
In[*]:= FullSimplify[EcoStableQ[eq[[2]]]]
```

```
Out[*]=
```

$$\begin{cases} \text{True} & r1 \alpha_{21} > r2 \alpha_{11} \\ \text{False} & r1 \alpha_{21} < r2 \alpha_{11} \mid \mid r1 (\alpha_{11} + \alpha_{21}) < r2 \alpha_{11} \\ \text{Indeterminate} & \text{True} \end{cases}$$

```
In[*]:= FullSimplify[EcoStableQ[eq[[3]]]]
```

```
Out[*]=
```

$$\begin{cases} \text{True} & r2 \alpha_{12} > r1 \alpha_{22} \\ \text{False} & r1 \alpha_{22} > r2 (\alpha_{12} + \alpha_{22}) \mid \mid r2 \alpha_{12} < r1 \alpha_{22} \\ \text{Indeterminate} & \text{True} \end{cases}$$

```
In[ ]:= FullSimplify[EcoStableQ[eq[[4]]]]
Out[ ]:=
```

$$\begin{cases} \text{True} & (\alpha_{12} \alpha_{21} - \alpha_{11} \alpha_{22}) (r_2 \alpha_{11} (\alpha_{12} - \alpha_{22}) + r_1 (-\alpha_{11} + \alpha_{21}) \alpha_{22}) > 0 \&\& \\ & (r_2 \alpha_{11} - r_1 \alpha_{21}) (r_2 \alpha_{12} - r_1 \alpha_{22}) (-\alpha_{12} \alpha_{21} + \alpha_{11} \alpha_{22}) < 0 \\ \text{False} & (\alpha_{12} \alpha_{21} - \alpha_{11} \alpha_{22}) (r_2 \alpha_{11} (\alpha_{12} - \alpha_{22}) + r_1 (-\alpha_{11} + \alpha_{21}) \alpha_{22}) < 0 \mid \mid \\ & (r_2 \alpha_{11} - r_1 \alpha_{21}) (r_2 \alpha_{12} - r_1 \alpha_{22}) (-\alpha_{12} \alpha_{21} + \alpha_{11} \alpha_{22}) > 0 \\ \text{Indeterminate} & \text{True} \end{cases}$$

Invasion rates:

```
In[ ]:= λ21 = Inv[eq[[2]], n2]
Out[ ]:=
```

$$r_2 - \frac{r_1 \alpha_{21}}{\alpha_{11}}$$

```
In[ ]:= λ12 = Inv[eq[[3]], n1]
Out[ ]:=
```

$$r_1 - \frac{r_2 \alpha_{12}}{\alpha_{22}}$$



---

## Numerical examples

### Case I - 1 outcompetes 2

```
In[ ]:= r1 = 1.2; r2 = 1;
α11 = 1; α22 = 1;
α12 = 1; α21 = 1;

In[ ]:= eq = SolveEcoEq[]
```

 **Solve:** Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result.

```
Out[ ]:=
```

$$\{\{n_1 \rightarrow 0, n_2 \rightarrow 0\}, \{n_1 \rightarrow 1.2, n_2 \rightarrow 0\}, \{n_1 \rightarrow 0, n_2 \rightarrow 1.\}\}$$

```
In[ ]:= EcoEigenvalues[eq]
Out[ ]:=
```

$$\{\{1.2, 1.\}, \{-1.2, -0.2\}, \{-1., 0.2\}\}$$

```
In[ ]:= NumberedGridForm[eq, EcoEigenvalues[eq], Header → True]
Out[ ]:=
```

#	eq	EcoEigenvalues[eq]
1	{n1 → 0, n2 → 0}	{1.2, 1.}
2	{n1 → 1.2, n2 → 0}	{-1.2, -0.2}
3	{n1 → 0, n2 → 1.}	{-1., 0.2}


```
In[ ]:= λ12 = Inv[eq[[3]], n1]
        λ21 = Inv[eq[[2]], n2]
```

```
Out[ ]:=
0.2
```

```
Out[ ]:=
-0.2
```


```
In[ ]:= pp = PlotEcoPhasePlane[{n1, 0, 1.5}, {n2, 0, 1.5}]
```

```
Out[ ]:=
```




```
In[ ]:= sol = EcoSim[{n1 -> 0.01, n2 -> 0.01}, 50];
        PlotDynamics[sol]
```

```
Out[ ]:=
```



```
In[ ]:= Show[pp, PlotTrajectory[sol, PlotStyle -> Black]]
Out[ ]:=
```



### Case III - 1 & 2 coexist

```
In[ ]:= r1 = 1; r2 = 1;
        α11 = 1; α22 = 1;
        α12 = 0.5; α21 = 0.5;

In[ ]:= eq = SolveEcoEq[];
        NumberedGridForm[eq, EcoEigenvalues[eq], Header -> True]
```

**Solve:** Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result.

```
Out[ ]:=
```

#	eq	EcoEigenvalues[eq]
1	{n1 -> 0, n2 -> 0}	{1., 1.}
2	{n1 -> 1., n2 -> 0}	{-1., 0.5}
3	{n1 -> 0, n2 -> 1.}	{-1., 0.5}
4	{n1 -> 0.666667, n2 -> 0.666667}	{-1., -0.333333}


```
In[ ]:= λ12 = Inv[eq[[3]], n1]
        λ21 = Inv[eq[[2]], n2]
```

```
Out[ ]:=
0.5
```

```
Out[ ]:=
0.5
```


```
In[ ]:= pp = PlotEcoPhasePlane[{n1, 0, 1.5}, {n2, 0, 1.5}]
```

```
Out[ ]:=
```




```
In[ ]:= sol = EcoSim[{n1 -> 0.02, n2 -> 0.01}, 50];
PlotDynamics[sol]
```

```
Out[ ]:=
```



```
In[ ]:= Show[pp, PlotTrajectory[sol, PlotStyle -> Black]]
Out[ ]:=
```



### Case IV - 1 or 2 wins (founder control)

```
In[ ]:= r1 = 1; r2 = 1;
        α11 = 1; α22 = 1;
        α12 = 2; α21 = 2;

In[ ]:= eq = SolveEcoEq[];
        NumberedGridForm[eq, EcoEigenvalues[eq], Header -> True]
```

```
Out[ ]:=
```

#	eq	EcoEigenvalues[eq]
1	{n1 → 0, n2 → 0}	{1, 1}
2	{n1 → 1, n2 → 0}	{-1, -1}
3	{n1 → 0, n2 → 1}	{-1, -1}
4	{n1 → $\frac{1}{3}$ , n2 → $\frac{1}{3}$ }	{-1, $\frac{1}{3}$ }


```
In[ ]:= λ12 = Inv[eq[[3]], n1]
        λ21 = Inv[eq[[2]], n2]
```

```
Out[ ]:=
-1
```

```
Out[ ]:=
-1
```


```
In[ ]:= pp = PlotEcoPhasePlane[{n1, 0, 1.2}, {n2, 0, 1.2}]
```

```
Out[ ]:=
```




```
In[ ]:= sol = EcoSim[{n1 -> 0.02, n2 -> 0.01}, 50];  
PlotDynamics[sol]
```

```
Out[ ]:=
```






```
In[ ]:= sol2 = EcoSim[{n1 → 0.01, n2 → 0.02}, 50];
PlotDynamics[sol2]
```



```
In[ ]:= Show[pp, PlotTrajectory[{sol, sol2}, PlotStyle → Black]]
```



## Case 0 - neutrality

```
In[ ]:= r1 = 1; r2 = 1;
α11 = 1; α22 = 1;
α12 = 1; α21 = 1;
```

```
In[ ]:= eq = SolveEcoEq[];
NumberedGridForm[eq, EcoEigenvalues[eq], Header → True]
```

**Solve:** Equations may not give solutions for all "solve" variables. ⓘ

```
Out[ ]:=
```


#	eq	EcoEigenvalues[eq]
1	$\{n1 \rightarrow 0, n2 \rightarrow 0\}$	$\{1, 1\}$
2	$\{n2 \rightarrow 1 - n1\}$	$\{-1, 0\}$

```
In[ ]:= λ12 = Inv[{n1 → 0, n2 → r2 / α22}, n1]
λ21 = Inv[{n1 → r1 / α11, n2 → 0}, n2]
```


```
Out[ ]:=
0
```

```
Out[ ]:=
0
```


```
In[ ]:= pp = PlotEcoPhasePlane[{n1, 0, 1.2}, {n2, 0, 1.2}]
```



```
In[ ]:= sol = EcoSim[{n1 → 0.02, n2 → 0.01}, 50];  
PlotDynamics[sol]
```



```
In[ ]:= sol2 = EcoSim[{n1 → 0.01, n2 → 0.02}, 50];  
PlotDynamics[sol2]
```



```
In[ ]:= Show[pp, PlotTrajectory[{sol, sol2}, PlotStyle -> Black]]  
Out[ ]:=
```

